



DECUS

PROGRAM LIBRARY

DECUS NO.	8-303
TITLE	ALTERATIONS OF THE BASIC FLOATING-POINT PACKAGE AND ADDITIONAL SUBROUTINES
AUTHOR	W. Roos
COMPANY	Philips Research Laboratories Eindhoven, Netherlands
DATE	March 18, 1970
SOURCE LANGUAGE	PAL

SECRET

CONFIDENTIAL

CONFIDENTIAL

CONFIDENTIAL

ALTERATIONS OF THE BASIC FLOATING-POINT PACKAGE AND ADDITIONAL SUBROUTINES

DECUS Program Library Write-up

DECUS No. 8-303

ABSTRACT

Using the basic floating-point package DEC-08-YQ1B or the other existing versions, this package can only be called from instruction- and data field zero. For a PDP-8/I with extended memory it becomes necessary for additional subroutines to overcome this drawback.

USAGE

Using the altered package with additional subroutines, there are the following restrictions:

1. Data- and instruction fields must be the same.
2. At the locations 5, 6 and 7 of page zero of each data field, the addresses of the additional subroutines are to be placed.

EX1	=	0040
AC1H	=	0041
AC1L	=	0042
OVER1	=	0043
EXP	=	0044
HORD	=	0045
LORD	=	0046
OVER2	=	0047
EXP1	=	0050
FLDV	=	6305
MINUS2	=	6400
TABLE6	=	6545
DMULT	=	6221
DNORM	=	6600
ALIGN	=	6020
DUNORM	=	6564

/ADDITIONAL SUBROUTINES OF FLOATING
 /POINT PACKAGE TO CALL F.P. FROM OTHER
 /FIELDS. INSTRUCTION AND DATA FIELD
 /MUST BE THE SAME.
 /SUBROUTINES CAN BE EASILY RELOCATED.

*0005

INPROU
 OUTROU
 AFLPR

*1401

AFLPR,	Ø	/ENTRY SUBROUTINE
	CLA CLL	
	RDF	/READ THE DATA FIELD
	CDF	
	JMS SFINS	/SAVE FIELD BITS, SET INSTR.
	TAD AFLPR	/GET ADDRESS OF NEXT INSTR.
	DCA I ADR1	/AND DEPOSIT AT ENTRY OF F.P.
	JMP I ADR2	/PACKAGE, ENTER F.P. PACKAGE
ADR1,	FPNT	
ADR2,	FPNT+1	
SFINS,	Ø	
	DCA OFLD1	/STORE THE FIELD BITS
	TAD OFLD1	
	TAD FLD	/A CDF INSTRUCTION
	DCA OFLD	/SET UP THE CDF INSTRUCTIONS
	TAD OFLD	/OF THE PROPER FIELD FROM WHICH
	DCA FLD1	/THE DATA IS TO BE TAKEN
	TAD OFLD	
	DCA FLD2	
	TAD OFLD	
	DCA FLD3	
	JMP I SFINS	
NEXI,	Ø	/SUBROUTINE TO GET THE
	DCA IFPNT	/NEXT F.P. INSTRUCTION
	DCA OVER1	
	DCA OVER2	
FLD1,	CDF	
	TAD I IFPNT	
	CDF	
	JMP I NEXI	
IFPNT,	Ø	
DNAD,	Ø	/SUBROUTINE TO GET THE
	DCA NADR	/MEMORY CONTENTS FROM THE
	JMS FNADR	/PROPER DATA FIELD AT
	JMP I DNAD	/THE SPECIFIED ADDRESS
NADR,	Ø	

GETF,	Ø	/SUBROUTINE TO GET THE
	DCA NADR	/MANTISSA FROM THE PROPER
	ISZ NADR	/DATA FIELD
	JMS FNADR	
	DCA AC1H	
	ISZ NADR	
	JMS FNADR	
	DCA AC1L	
	JMP I GETF	
FNADR,	Ø	
FLD2,	CDF	
	TAD I NADR	
	CDF	
	JMP I FNADR	
DFLPT,	Ø	/SUBROUTINE TO PUT THE
	DCA NADR	/F.A. AT SPECIFIED ADDRESS
	TAD EXP	/AND DATA FIELD
	JMS FPIADR	
	ISZ NADR	
	TAD HORD	
	JMS FPIADR	
	ISZ NADR	
	TAD LORD	
	JMS FPIADR	
	JMP I DFLPT	
FPIADR,	Ø	
FLD3,	CDF	
	DCA I NADR	
	CDF	
	JMP I FPIADR	
DEXI,	CLA CLL	/EXIT ROUTINE
	TAD I ADR1	/GET RETURN ADDRESS
	DCA AFLPR	
	TAD OFLD	/A CDF INSTRUCTION
	DCA NINST1	
	TAD OFLD1	/GET THE FIELD BITS
	TAD OCIF	/A CIF INSTRUCTION
	DCA NINST2	
NINST1,	CDF	
NINST2,	CIF	
	JMP I AFLPR	/RETURN
OFLD,	CDF	
OCIF,	CIF	
FLD,	CDF	
OFLD1,	CDF	
OFLD2,	Ø	

```

DSFLD, 0
      TAD OFLD1      /SAVE FIELD BITS
      DCA OFLD2
      JMP I DSFLD
DRFLD, 0
      TAD OFLD2      /SET ORIGINAL FIELD INSTR.
      JMS SFINS
      JMP I DRFLD
INPROU, 0      /ADDITIONAL INPUT SUBR.
      CLA CLL
      RDF
      CDF
      DCA NFLD1      /STORE THE FIELD BITS
      JMS I INPROU1  /INPUT SUBR. OF F.P.
      CLA CLL        /SET UP THE CDF AND
      TAD NFLD1      /CIF INSTRUCTIONS
      TAD FLD
      DCA NINST3
      TAD NFLD1
      TAD OCIF
      DCA NINST4
NINST3, CDF
NINST4, CIF
      JMP I INPROU   /RETURN
INPROU1, 7400
OUTROU, 0      /ADDITIONAL OUTPUT SUBR.
      CLA CLL
      RDF
      CDF
      DCA NFLD1      /STORE THE FIELD BITS
      JMS I OUTROU1 /OUTPUT SUBR. OF F.P.
      CLA CLL        /SET UP THE CDF AND
      TAD NFLD1      /CIF INSTRUCTIONS
      TAD FLD
      DCA NINST5
      TAD NFLD1
      TAD OCIF
      DCA NINST6
NINST5, CDF
NINST6, CIF
      JMP I OUTROU   /RETURN
OUTROU1, 7200
NFLD1, 0

```

/ALTERATIONS OF FIRST PAGE OF
 /FLOATING POINT PACKAGE

*5600

```

FPNT,      0
           CLA CLL
           TAD FPNT
           JMS I INEXT      /GET NEXT INSTRUCTION
           DCA JUMP
           TAD JUMP
           AND PAGENO      /GET PAGE BIT
           SNA CLA         /PAGE ZERO?
           JMP .+3         /YES
           TAD MASK5      /NO
           AND FPNT       /C(FPNT) 0-4 CONTAINS PAGE BITS
           DCA ADDR
           TAD MASK7      /GET 7-BIT ADDRESS
           AND JUMP
           TAD ADDR
           DCA ADDR
           TAD INDRCT     /INDIRECT BIT=1?
           AND JUMP
           SNA CLA
           JMP LOOP01     /NO - GO ON
           TAD ADDR       /YES DEFER
           JMS I IDNAD
           DCA ADDR
LOOP01,    ISZ FPNT
           TAD ADDR
           JMS I IDNAD
           DCA EX1        /EXPONENT
           TAD ADDR
           JMS I IGETF     /GET MANTISSA
           TAD JUMP
           CLL RTL
           RTL
           AND MASK3      /GET BITS 0-2, IE OPCODE
           TAD TABLE     /LOOKUP IN TABLE
           DCA JUMP2
           TAD I JUMP2
           DCA JUMP2
           JMP I JUMP2     /GO THERE

JUMP,      0
JUMP2,     0
GO2,       0
ADDR,      0
MASK3,     0017
MASK5,     7600
MASK7,     0177
PAGENO,    0200
INDRCT,    0400
  
```

TABLE,	.+1	/TABLE USED IN INTERPRETING
	EXIT	/BITS 0-2 OF PSEUDO
	FLAD	/INSTRUCTION
	FLSU	/IF OPCODE=0 GO TO EXIT
	FLMY	/AND INTERPRET BITS 8-11
	FLDV	
	FLGT	
	FLPT	
	NORF	
FLGT,	TAD EX1	/FGET=5
	DCA EXP	
	TAD AC1H	
	DCA HORD	
	TAD AC1L	
	DCA LORD	
	JMP FPNT+1	
FLPT,	TAD ADDR	/FPUT=6
	JMS I IFLPT	/DEPOSIT SUBROUTINE
	JMP FPNT+1	
FLAD,	JMS I ALGN	/FLAD=1, FIRST ALIGN
	JMP FPNT+1	/EXPONENTS, RETURN IF NO
	JMS I UNORM	/ALIGNMENT IS POSSIBLE
	CLL	/LARGER OF THE TWO IS IN F.A.
	TAD OVER1	/TRIPLE PRECISION ADDITION
	TAD OVER2	/SINCE BITS ARE SHIFTED
	DCA OVER2	/RIGHT
	RAL	
	TAD AC1L	
	TAD LORD	
	DCA LORD	
	RAL	
	TAD AC1H	
	TAD HORD	
	DCA HORD	
	JMS I NORM	/NORMALIZE THE RESULT
	JMP FPNT+1	
FLSU,	JMS I OPMINS	/FSUB=2, NEGATE THE OPERAND
	JMP FLAD	/ADD
OPMINS,	MINUS2	
EXIT,	TAD JUMP	/OPCODE=0
	AND MASK3	/ARE BITS 8-11=0?
	SNA	
	JMP I IEXI	/YES, GO TO EXIT ROUTINE
	TAD ACON6	/LOOKUP ON TABLE
	DCA JUMP2	
	TAD I JUMP2	
	DCA JUMP2	
	TAD FPNT	

	DCA G02	
	JMS I ISFLD	/SAVE FIELD BITS
	JMS I JUMP2	/CALL AS A SUBROUTINE
	JMS I IRFLD	/RESTORE FIELD INSTR.
	TAD G02	/RESTORE F.P. POINTER
	DCA FPNT	
	JMP FPNT+1	/GET NEXT PSEUDO INSTRUCTION
ACON6,	TABLE6-1	/CALLING CAN BE TO A DEPTH ONE
FLMY,	CLA IAC	/FMPY=3
	TAD EX1	
	TAD EXP	/ADD EXPONENTS TOGETHER
	DCA EXP	
	JMS I MULT	/MULTIPLY
	JMP FPNT+1	
MULT,	DMULT	
NORM,	DNORM	
ALGN,	ALIGN	
UNORM,	DUNORM	
NORF,	JMS I NORM	
	JMP FPNT+1	
INEXT,	NEXI	
IDNAD,	DNAD	
IGETF,	GETF	
IFLPT,	DFLPT	
IEXI,	DEXI	
ISFLD,	DSFLD	
IRFLD,	DRFLD	
*6335		
	FLAD+2	/ALTERATION JMP INSTRUCTION

